

Retracing My SAS® Global Forum 2016 Steps: Visualizing iPhone Health App Step Data With a (What Else?) Step Plot

Ted Conway, Chicago, IL

ABSTRACT

If you've got an iPhone, you might have noticed that the Health app is hard at work collecting data on every step you take. And, of course, the data scientist inside you is itching to analyze that data with SAS®. This paper and an accompanying E-Poster show you how to get step data out of your iPhone Health app and into SAS. Once it's there, you can have at it with all things SAS. In this presentation, we show you how a (what else?) step plot can be used to visualize the 73,000+ steps the author took at SAS® Global Forum 2016.

INTRODUCTION

Several months after returning from SAS Global Forum 2016 in Las Vegas, I opened my iPhone's Health App for the first time and was surprised to learn it had been dutifully tracking the steps I'd taken since the day I bought it, including those during SGF (so much for "what happens in Vegas stays in Vegas!").

And so, like any good SAS user, I felt compelled to explore the newly-discovered data with SAS!

GETTING THE DATA

My quest for the iPhone Health App step data began with a Google search for "iphone health app download data." The search results were not particularly encouraging:

"There are actually two ways to export the data from your Health app," began one article. "The first way, is one provided by Apple, but it is virtually useless. [...] this way of exporting your data is pretty much useless because the data is then presented in an xml file, making it hard to do anything with the data. It's not like you can simply open up the file and look at it."

But, as things turn out, it's nothing SAS's XML LIBNAME can't handle!




Fig. 1: iPhone Health App Export Feature

You can just email yourself a .zip file of the step data using the Health App's Export feature, uncompress the data, and read the export.xml file into SAS (sample code in **Appendix**).

SHOWING WHERE YOU'VE BEEN

With apologies to Edward Tufte, when it comes to presenting step data, even the Microsoft and Google Map Brainiacs seem to concede it's hard to beat the work of Bill Keane, whose old-school dotted black lines simply and eloquently conveyed the meanderings of little Billy and Jeffy in *The Family Circus*.




Fig. 2: © Family Circus: © 2015 Bill Keane, Inc. Distributed by King Features Syndicate. Reprinted with permission.

At SGF 2016, my not-unlike-Billy's wanderings were largely confined to the area of the Las Vegas Strip shown in the following Bing Maps screenshot. This included my hotel (Flamingo, lower left), the convention center (Sands Expo, upper right), and the conference hotel (The Venetian, middle).



Fig. 3: Bing Maps Walking Directions

Unfortunately, iPhone Health App step data lacks the precision that would be needed to try to do a literal Family Circus or Google/Bing Map-like take on the data.

So, what might be the next best way of presenting the 73,000+ steps taken over four days at SGF, keeping with the spirit of *The Family Circus*?

A **step plot**, of course!

STEPPIN' OUT




Fig. 4: ODS Graphics Step Plot of Step Data (Full Code & Larger Image in Appendix)

While perhaps not obvious at first glance to an observer, the step plot did a surprisingly good job of helping me retrace my steps during the four days spent at the conference:

- On Monday and Thursday, the longest flat lines represent the hours spent flying from Chicago (ORD) to Vegas (LAS) and back again.
- The steeper ascents of lines on Monday thru Thursday represent time spent almost continuously walking – the lengthy walks between my hotel (Flamingo) and the conference hotel (Venetian), a side trip over to Caesars on Tuesday (to sadly see the Blackhawks lose), sightseeing, and circulating at the Quad and other networking events.
- Other plateaus represent time spent sitting/standing for various lengths – Opening Session, Hands-On Workshops, Keynotes, shorter sessions and Quad demos, subway and Uber rides to and from the airport, meals. OK, and maybe a catnap or two!

While somewhat less code would have been required for a “no-frills” step plot that still would have gotten the job done, this was for a 40” x 22” SGF 2017 e-poster, so a few extras were tossed in to (hopefully!) increase visual interest.

- Text annotation features were used to add prominent headings for each day of the conference.
- Image annotation features were employed to add photos taken at the event, and text annotation was used to add captions.
- Reference lines were added to show hourly grid lines with Pacific Daylight Time (PDT) labels.
- SAS’s broken-axis features were used to recapture lots of screen real estate that would have been otherwise lost to long flat lines at the top of the chart representing the hours spent sleeping on Monday-Tuesday-Wednesday nights, when no steps were taken. A cue to how many hours were spent sleeping each night can still be found in the varying widths of each “logical” conference day shown in the plot.

CONCLUSION

Getting data out of the iPhone Health app and into SAS opens the door to a wealth of data exploration and visualization opportunities.

It's just what the Doctor ordered for casual do-it-yourself data scientists and die-hard members of the quantified-self movement!

ACKNOWLEDGMENTS AND RECOMMENDED READING

SAS. *Base SAS® Output Delivery System (ODS) Graphics Suite*

<http://support.sas.com/documentation/prod-p/grstat/index.html>

SAS. *XML LIBNAME Engine: User's Guide*

<http://support.sas.com/documentation/cdl/en/engxml/64990/HTML/default/viewer.htm>

SAS. *Graphically Speaking: Data Visualization with a focus on SAS ODS Graphics*

<http://blogs.sas.com/content/graphicallyspeaking/>

Conners, Shannon. *Digging into my diet and fitness data*

<https://community.jmp.com/t5/JMP-Blog/Digging-into-my-diet-and-fitness-data/ba-p/30474>

Page, Sébastien. *How to export and import your Health data*

<http://www.idownloadblog.com/2015/06/10/how-to-export-import-health-data/>

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Ted Conway

ted.j.conway@gmail.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.

APPENDIX I – SAS CODE

```

* Step plot of steps tracked by iPhone Health App at SAS Global Forum 2016
See idownloadblog.com/2015/06/10/how-to-export-import-health-data/ for how to get step data;

filename health '/folders/myfolders/export.xml'; * XML data provided by iPhone Health App;
filename map '/folders/myfolders/HealthData.map';
libname health xmlv2 xmlmap=map automap=replace;

proc sql;
create table steps as
select input(substr(record_startDate,1,19),YMDT19.) as startDT format=datetime26.,
       input(substr(record_endDate,1,19),YMDT19.) as endDT format=datetime26., record_value as steps
from health.Record where (record_type="HKQuantityTypeIdentifierStepCount")
where record_startDate between '2016-04-18' and '2016-04-22 05' order by 1;

data totSteps(keep=time totsteps);
retain totsteps 0;
set steps end=eof;
if intck('hour',lag1(enddt),startdt)>4 then do;
    put "End of Day, Steps Taken=: " totsteps;
    totsteps=0;
end;
time=startdt;
output;
totsteps+steps;
time=enddt;
format time datetime19.;
output;
if eof;
put "End of Day, Steps Taken=: " totsteps;

options macrogen;
data anno;
length image label $100. anchor widthunit textcolor function x1space y1space border textweight $20.;

function="text"; y1space='datavalue'; x1space='datavalue';
border='false'; textcolor="GRAY"; anchor="bottomleft";
width=17; widthunit="percent"; textsize=38; textweight="bold"; y1=24250;
x1='18APR2016:05:20:00'dt; label="MONDAY APRIL 18 27,621 STEPS"; output;
x1='19APR2016:10:00:00'dt; label="TUESDAY APRIL 19 16,956 STEPS"; output;
x1='20APR2016:09:45:00'dt; label="WEDNESDAY APRIL 20 15,749 STEPS"; output;
x1='21APR2016:10:00:00'dt; label="THURSDAY APRIL 21 13,112 STEPS"; output;

* Macro to insert captioned photos taken at SGF 2016;
%macro photo(lbl=, img=, time=, steps=, width=);
function="Image";
image="/folders/myfolders/SGF2017Photos/&img"; anchor="bottomleft";
imagescale='fitwidth'; layer='back'; border='true'; linethickness=10; linecolor="black";
drawspace="wallpercent"; x1="&time"dt; y1=&steps; width=&width; widthunit='percent';
x1space='datavalue'; y1space='datavalue';
output;
function="text";
label="&lbl"; anchor="topleft"; border='false'; textcolor="BLACK";
width=20; widthunit="percent"; textsize=14; textweight="bold";
x1space='datavalue'; y1=y1-100; y1space='datavalue';
output;
%mend;

* Day 1 photos;
%photo(lbl=VENETIAN LOBBY, img=VenetianLobby.jpg, time=18APR2016:19:45:00, steps=7000, width=9.41);
%photo(lbl=VENETIAN SHOPS, img=VenetianShopsV2.jpg, time=18APR2016:05:45:00, steps=18250, width=17.42);
%photo(lbl=ORD TO LAS, img=SpiritOrdToLAS.jpg, time=18APR2016:05:45:00, steps=7000, width=10.53);
%photo(lbl=CTA BLUE LINE, img=BlueLine.jpg, time=18APR2016:14:00:00, steps=1000, width=12.41);
%photo(lbl=FLAMINGO ROOM VIEW, img=ViewFlamingoRoom.jpg, time=18APR2016:05:45:00, steps=11600, width=12.30);
%photo(lbl=SGF MINGLE, img=SgfMonEveMingle.jpg, time=18APR2016:22:30:00, steps=15350, width=6.30);

```

Fig. 5: SAS Code to read and plot iPhone Health app steps data

APPENDIX I – SAS CODE (CONT.)

```

* Day 2 photos;
%photo(lbl=E-MINER HANDS-ON, img=SasEmHandsOn.jpg, time=19APR2016:10:30:00, steps=6750, width=9.38);
%photo(lbl=SAS STUDIO HANDS-ON, img=SasStudioStatHandsOn.jpg, time=19APR2016:17:30:00, steps=1000,
width=8.95);
%photo(lbl=WELCOME RECEPTION (QUAD), img=QuadWhatsNewWithSGProcs.jpg, time=19APR2016:10:30:00, steps=19800,
width=18.09);
%photo(lbl=OPENING SESSION, img=OpeningSession.jpg, time=19APR2016:10:30:00, steps=11750, width=15.87);

* Day 3 photos;
%photo(lbl=TECHNOLOGY CONNECTION, img=TechConnection.jpg, time=20APR2016:10:15:00, steps=18150, width=21.12);
%photo(lbl=KICKBACK PARTY, img=BrooklynBowlKickback.jpg, time=20APR2016:10:15:00, steps=9100, width=10.45);
%photo(lbl=REDEFINING SUCCESS, img=AriannaHuffington.jpg, time=20APR2016:19:30:00, steps=1000, width=8.50);
%photo(lbl=DATA @FORD, img=Ford.jpg, time=20APR2016:18:15:00, steps=13500, width=6.54);
%photo(lbl=VIYA, img=Viya.jpg, time=20APR2016:13:45:00, steps=1000, width=6.46);

* Day 4 photos;
%photo(lbl=THE SAS PACK, img=SasPack.jpg, time=21APR2016:18:30:00, steps=1000, width=9.65);
%photo(lbl=GTL HANDS-ON, img=GtlHandsOn.jpg, time=21APR2016:10:30:00, steps=14050, width=21.68);
%photo(lbl=EXCEL PLAYGROUND, img=ExcelPlayground.jpg, time=21APR2016:10:30:00, steps=9700, width=9.15);
%photo(lbl=CASINO, img=Slots.jpg, time=21APR2016:10:30:00, steps=20250, width=4.60);
%photo(lbl=B&B BURGER, img=BANDBBurger.jpg, time=21APR2016:14:25:00, steps=20250, width=7.41);
%photo(lbl=LAS TO ORD, img=LASToORD.jpg, time=21APR2016:20:15:00, steps=20250, width=7.45);

data _null_;
length reflines labels $ 4096;
do d='18APR2016:00:00:00'dt to '21APR2016:00:00:00'dt by 60*60*24;
    do h=6 to 27;
        t=d+h*60*60;
        reflines=trim(reflines)||" "||put(t,datetime16.)||"dt";
        l=t-2*60*60;
        daylabel='';
        if 1 in ('18APR2016:04:00:00'dt '19APR2016:08:00:00'dt '20APR2016:08:00:00'dt '21APR2016:08:00:00'dt')
            then daylabel='*'||upcase(put(datepart(1),downcase3.));
        labels=trim(labels)||" "||compress(put(timepart(1),timeampm5.))||trim(daylabel)||"";
    end;
end;
call symput('reflines', reflines);
call symput('labels', labels);
run;
%put &labels;

ods all close;
ods listing image_dpi=150 gpath='/folders/myfolders' device=png;
ods graphics on / reset width=40in height=22in imagename="SGF2017steps" imagefmt=png antialias;

proc sgplot data=totsteps sganno=anno nowall;
step x=time y=totsteps / lineattrs=(thickness=3pt);
yaxis display=(nolabel) valueattrs=(size=8.75pt weight=bold) grid gridattrs=(pattern=dot color=black)
values=(0 to 28000 by 1000) valuesformat=comma9. grid;
xaxis display=(nolabel) type=time /* Broken x-axis for sleeping hours (no steps) */
ranges=('18APR2016:05:21:41'dt-'19APR2016:03:15:37'dt
'19APR2016:09:49:46'dt-'20APR2016:00:35:03'dt
'20APR2016:09:40:54'dt-'21APR2016:01:47:23'dt
'21APR2016:09:48:48'dt-'22APR2016:01:55:52'dt')
values=('18APR2016:05:21:41'dt to "22APR2016:01:55:52'dt by "06:00:00"t);
refline &reflines /
    lineattrs=(pattern=dot color=black) labelattrs=(size=8.75pt weight=bold) axis=x splitter='*'
    label=(&labels) labelloc=outside labelpos=min;
footnote height=32pt bold "Retracing My SAS® Global Forum 2016 Steps: Visualizing iPhone Health App Step Data
With a (What Else?) Step Plot";

```

Fig. 5: SAS Code to read and plot iPhone Health app steps data (cont.)

APPENDIX II – OPTIONAL XML MAP (UNNEEDED IF AUTOMAP SPECIFIED IN LIBNAME)

```
<?xml version="1.0" ?>
<SXLEMAP version="1.2">
  <TABLE name="RECORDS">
    <TABLE-PATH syntax="XPATH">/HealthData/Record</TABLE-PATH>
    <COLUMN name="startDate"><PATH>/HealthData/Record@startDate</PATH><TYPE>character</TYPE>
    <DATATYPE>STRING</DATATYPE><LENGTH>40</LENGTH></COLUMN>
    <COLUMN name="endDate"><PATH>/HealthData/Record@endDate</PATH><TYPE>character</TYPE>
    <DATATYPE>STRING</DATATYPE><LENGTH>40</LENGTH></COLUMN>
    <COLUMN name="type"><PATH>/HealthData/Record@type</PATH><TYPE>character</TYPE>
    <DATATYPE>STRING</DATATYPE><LENGTH>100</LENGTH></COLUMN>
    <COLUMN name="unit"><PATH>/HealthData/Record@unit</PATH><TYPE>character</TYPE>
    <DATATYPE>STRING</DATATYPE><LENGTH>40</LENGTH></COLUMN>
    <COLUMN name="value"><PATH>/HealthData/Record@value</PATH><TYPE>numeric</TYPE>
    <DATATYPE>integer</DATATYPE></COLUMN>
  </TABLE>
</SXLEMAP>
```

Fig. 6: Hand-coded HealhData.map (Optional. SAS generates an XML map if AUTOMAP is specified)

APPENDIX III – OUTPUT (40" X 22" .PNG IMAGE FOR SGF 2017 E-POSTER)




Fig. 7: ODS Graphics Step Plot of Steps Taken at SGF 2016 (Smaller Version of 40"x22" e-Poster)